

# Paraxial optics : An example -Cemented doublet-


Dr.TANAKA, Kazuo

E-mail address : tanaka.kazuo205@canon.co.jp, User494704@aol.com

October 17, 2000

近軸光学の簡単な計算例として、「貼合せ単レンズ」を取り上げる。

## Cross sectional view of an example



```
> restart: with(linalg): Digits:=25:
```

```
Warning, the protected names norm and trace have been redefined and unprotected
```

### ABCD Gaussian constants by means of Gaussian brackets

まず、ABCDガウス係数をガウシアンブラケットで定義しておく。

```
> AA:=proc(m,n) -CC(m,n-1)*e(n-1)+AA(m,n-1) end: AA(m,m ):=1:
BB:=proc(m,n) -DD(m,n-1)*e(n-1)+BB(m,n-1) end: BB(m,m ):=0:
CC:=proc(m,n) AA(m,n)*phi(n)+CC(m,n-1) end: CC(m,m-1):=0:
DD:=proc(m,n) BB(m,n)*phi(n)+DD(m,n-1) end: DD(m,m ):=1:
```

### Initial data

ここで初期データ及びそれらに纏わる諸量を与える。

#### Lens data & Object distance

ここで、数値例のレンズデータ（曲率半径、レンズ厚、屈折率）及び物体距離を与える。  
このパラグラフを飛ばして以下に行くと数値解でなく、解析解が得られる。  
また、数値から小数点を削除すると整数表示の解、小数点を付加すると少数表示の解を得る。

#### Lens data

曲率半径、レンズ厚、屈折率

```
> r[1]:=29: r[2]:=-31: r[3]:=-127:
d[1]:=7: d[2]:=2:
n[0]:=1301/1031: n[1]:=3001/2003: n[2]:=9013/5003:
n[3]:=1109/1009:
```

#### Object distance (actual value)

物体距離

```
> ObjectDistance:=-3607:
```

#### Reduce distances & Powers

[ 換算間隔（光軸方向の実寸法距離を屈折率で除した量）及び各面の屈折力を求める。  
 前パラグラフ及び本パラグラフを飛ばして次に進むと屈折力と換算間隔で表す解析解を得る。

### Reduced distances

[ 換算間隔

```
> e[0]:=-ObjectDistance/n[0]; e[1]:=d[1]/n[1]; e[2]:=d[2]/n[2];
```

$$e_0 := \frac{3718817}{1301}$$

$$e_1 := \frac{14021}{3001}$$

$$e_2 := \frac{10006}{9013}$$

### Powers

[ 屈折力

```
> phi[1]:=(n[1]-n[0])/r[1]; phi[2]:=(n[2]-n[1])/r[2];
   phi[3]:=(n[3]-n[2])/r[3];
```

$$\phi_1 := \frac{16832}{2065093}$$

$$\phi_2 := \frac{-3039036}{310651279}$$

$$\phi_3 := \frac{3545790}{641099429}$$

### ABCD Gaussian constants

[ ABCDガウス係数を2つの方法で求める。

#### ABCD Gaussian constants

##### Gaussian constants by means of paraxial ray tracing

[ 初期値を  $\alpha'_0 = 0, h_1 = 1$  として近軸光線追跡を行うと  $\alpha'_{last} = C, h_{last} = A$ 、

```
> alpha[0]:=0; h[1]:=1;
   alpha[1]:=alpha[0]+h[1]*phi[1]; h[2]:=h[1]-alpha[1]*e[1];
   alpha[2]:=alpha[1]+h[2]*phi[2]; h[3]:=h[2]-alpha[2]*e[2];
   alpha[3]:=alpha[2]+h[3]*phi[3];
```

$$\alpha_0 := 0$$

$$h_1 := 1$$

```
> a[RayTrace]:=h[3]; c[RayTrace]:=alpha[3];
```

$$a_{RayTrace} := \frac{1668038239461415}{1731556531616479}$$

$$c_{RayTrace} := \frac{4516311713281087658566}{1110099903700545133890491}$$

[ そして、初期値を  $\alpha'_0 = 1, h_1 = 0$  として近軸光線追跡を行うと  $\alpha'_{last} = D, h_{last} = B$  を得る。

```
> alpha[0]:=1; h[1]:=0;
   alpha[1]:=alpha[0]+h[1]*phi[1];
   h[2]:=h[1]-alpha[1]*e[1];
   alpha[2]:=alpha[1]+h[2]*phi[2];
   h[3]:=h[2]-alpha[2]*e[2];
   alpha[3]:=alpha[2]+h[3]*phi[3];
```

$$\alpha_0 := 1$$

$$h_1 := 0$$

```
> b[RayTrace]:=h[3]; d[RayTrace]:=alpha[3];
```

$$b_{RayTrace} := \frac{-4890924153}{838488403}$$

$$d_{RayTrace} := \frac{544781849651583485}{537554436386421887}$$

#### Gaussian constants by means of Gaussian brackets

漸化式で与えたガウシアンブラケットによりA,B,C及びDを求める.

```
> AA(m,m+2):
a[GaussianBrackets]:=subs(m=1,phi(1)=phi[1],phi(2)=phi[2],phi
i(3)=phi[3],e(1)=e[1],e(2)=e[2],%);
BB(m,m+2):
b[GaussianBrackets]:=subs(m=1,phi(1)=phi[1],phi(2)=phi[2],ph
i(3)=phi[3],e(1)=e[1],e(2)=e[2],%);
CC(m,m+2):
c[GaussianBrackets]:=subs(m=1,phi(1)=phi[1],phi(2)=phi[2],ph
i(3)=phi[3],e(1)=e[1],e(2)=e[2],%);
DD(m,m+2):
d[GaussianBrackets]:=subs(m=1,phi(1)=phi[1],phi(2)=phi[2],ph
i(3)=phi[3],e(1)=e[1],e(2)=e[2],%);
```

$$a_{GaussianBrackets} := \frac{1668038239461415}{1731556531616479}$$

$$b_{GaussianBrackets} := \frac{-4890924153}{838488403}$$

$$c_{GaussianBrackets} := \frac{4516311713281087658566}{1110099903700545133890491}$$

$$d_{GaussianBrackets} := \frac{544781849651583485}{537554436386421887}$$

#### Check

2つの方法で求めたABCDガウス係数が等しいことを確認する.

```
> a[RayTrace]-a[GaussianBrackets];
b[RayTrace]-b[GaussianBrackets];
c[RayTrace]-c[GaussianBrackets];
d[RayTrace]-d[GaussianBrackets];
0
0
0
0
```

#### Lagrange-Helmholtz invariant

求めたABCDガウス係数を要素とするABCD行列の行列式の値が1であること、すなわち、Lagrange-Helmholtz不変量の成立を確認する.

```
> matrix([[a[RayTrace], b[RayTrace]], [c[RayTrace],
d[RayTrace]]]);
```

$$\begin{bmatrix} \frac{1668038239461415}{1731556531616479} & \frac{-4890924153}{838488403} \\ \frac{4516311713281087658566}{1110099903700545133890491} & \frac{544781849651583485}{537554436386421887} \end{bmatrix}$$

```
> Lagrange_Helmholtz_Invariant:=det(%);
```

$$Lagrange\_Helmholtz\_Invariant := 1$$

#### Imaging based on each surface data

##### Lens back (reduced value)

既述の物体距離及びレンズ系より、像位置を与えるレンズバック(最終面から像面までの距離)を2つの方法で求める.

## Lens back

### Lens back by means of paraxial ray tracing

[ まず、近軸光線追跡でレンズバックを求める。

```
> alpha[0]:=-1/e[0]; h[1]:=1;  
alpha[1]:=alpha[0]+h[1]*phi[1]; h[2]:=h[1]-alpha[1]*e[1];  
alpha[2]:=alpha[1]+h[2]*phi[2]; h[3]:=h[2]-alpha[2]*e[2];  
alpha[3]:=alpha[2]+h[3]*phi[3];
```

$$\alpha_0 := \frac{-1301}{3718817}$$

$$h_1 := 1$$

```
> s[kd_RayTrace]:=h[3]/alpha[3];
```

$$s_{kd\_RayTrace} := \frac{3865418742702576549842534456}{14870687693452272826135607}$$

### Lens back by means of Gaussian brackets

[ ABCDガウス係数によりレンズバックを求める。

```
> BB(m,m+3)/DD(m,m+3):  
s[kd_GaussianBrackets]:=subs(m=0,phi(1)=phi[1],phi(2)=phi[2],  
phi(3)=phi[3],
```

```
e(0)=e[0],e(1)=e[1],e(2)=e[2],%);
```

$$s_{kd\_GaussianBrackets} := \frac{3865418742702576549842534456}{14870687693452272826135607}$$

## Check

[ 両方法で求めたレンズバックが等しいことを確認する。

```
> s[kd_RayTrace]-s[kd_GaussianBrackets]: normal(%);  
0
```

## Lateral magnification

[ 上記の結像関係の横倍率を求める。横倍率は物高と像高の比で定義される。  
また、変形すると横倍率は射出換算傾角と入射換算傾角の比となる。  
ここで、近軸光線追跡とABCDガウス係数との方法で横倍率を求める。

## Lateral magnification

### Lateral magnification by means of paraxial ray tracing

[ 近軸光線追跡により横倍率を求める。

#### Ratio of object height to image height

[ 初期値を  $\alpha_0 = 0, h_1 = 1$   
として近軸光線追跡を行い物高と像高との比より横倍率を求める。

```
> alpha[0]:=0; h[1]:=1;  
alpha[1]:=alpha[0]+h[1]*phi[1];  
h[2]:=h[1]-alpha[1]*e[1];  
alpha[2]:=alpha[1]+h[2]*phi[2];  
h[3]:=h[2]-alpha[2]*e[2];  
alpha[3]:=alpha[2]+h[3]*phi[3];  
h[4]:=h[3]-alpha[3]*s[kd_RayTrace];
```

$$\alpha_0 := 0$$

$$h_1 := 1$$

```
> LateralMagnification[RayTrace_Height]:=h[4]/h[1];
```

$$LateralMagnification_{RayTrace\_Height} := \frac{-1400814718442685954598961}{14870687693452272826135607}$$

#### Ratio of emerging reduced inclination angle to incident reduced inclination

## angle

初期値を  $\alpha_0 = -\frac{1}{e'_0}, h_1 = 1$   
として近軸光線追跡を行い射出換算傾角と入射換算傾角との比より横倍率を求める。

```
> alpha[0]:=-1/e[0]; h[1]:=1;
alpha[1]:=alpha[0]+h[1]*phi[1]:
h[2]:=h[1]-alpha[1]*e[1]:
alpha[2]:=alpha[1]+h[2]*phi[2]:
h[3]:=h[2]-alpha[2]*e[2]:
alpha[3]:=alpha[2]+h[3]*phi[3]:

alpha_0 := -1301
          3718817
h_1 := 1

> LateralMagnification[RayTrace_Angle]:=alpha[0]/
alpha[3];

LateralMagnification_RayTrace_Angle := -1400814718442685954598961
                                         14870687693452272826135607
```

## Lateral magnification by means of Gaussian brackets

ABCDガウス係数により横倍率を求める。

```
> AA(m,m+3):
LateralMagnification[GaussianBrackets_A]:=subs(m=1,phi(1)=ph
i[1],phi(2)=phi[2],phi(3)=phi[3],

e(1)=e[1],e(2)=e[2],e(3)=s[kd_RayTrace],%);
1/DD(m,m+3):
LateralMagnification[GaussianBrackets_D]:=subs(m=0,phi(1)=ph
i[1],phi(2)=phi[2],phi(3)=phi[3],
e(0)=e[0],e(1)=e[1],e(2)=e[2],%);

LateralMagnification_GaussianBrackets_A := -1400814718442685954598961
                                             14870687693452272826135607

LateralMagnification_GaussianBrackets_D := -1400814718442685954598961
                                             14870687693452272826135607
```

## Check

両方法で求めた横倍率が等しいことを確認する。

```
> LateralMagnification[RayTrace_Height]-LateralMagnification[RayTr
ace_Angle]:
normal(%);
LateralMagnification[GaussianBrackets_A]-LateralMagnification[Ga
ussianBrackets_D]:
normal(%);
LateralMagnification[RayTrace_Height]-LateralMagnification[Gauss
ianBrackets_A]:
normal(%);

0
0
0
```

## Imaging based on the principal points

ここで、主点位置を基準とした結像を考察する。主点は横倍率が+1となる共役点である。物体頂点から出て前側主点へ向かう光線は後側主点より出て像頂点へ向かい、かつ、入射光線と射出光線の換算傾角は等しい。従って、横倍率は幾何学的な考察、三角形の相似関係、より次式で与えられる。

$$\text{Lateral magnification} = -\frac{s'_K - s'_P}{e'_0 + s_P}$$

## Imaging based on the principal points

### Principal points

まず、主点位置を求める。

#### Positions of principal points

[ 前側主点位置(第1面より前側主点までの距離)及び後側主点位置(最終面より後側主点までの距離)をそれぞれABCDガウス定数を用いて求める。

```
> s[p]:=(1-d[GaussianBrackets])/c[GaussianBrackets];  
s[pd]:=(a[GaussianBrackets]-1)/c[GaussianBrackets];
```

$$s_p := \frac{-7462640270996179949307}{2258155856640543829283}$$

$$s_{pd} := \frac{-20360770415833354929228}{2258155856640543829283}$$

#### check

主点が満足すべき条件を確認する。

#### Conjugate relation

[ 前側主点を物体そして後側主点を像とする結像関係を確認する。

```
> BB(m,m+4):  
subs(m=0,phi(1)=phi[1],phi(2)=phi[2],phi(3)=phi[3],e(0)=-  
s[p],e(1)=e[1],e(2)=e[2],e(3)=s[pd],%):  
normal(%);
```

0

#### Lateral magnification is unity

[ 横倍率が+1であることを確認する。

```
> AA(m,m+3):  
subs(m=1,phi(1)=phi[1],phi(2)=phi[2],phi(3)=phi[3],e(1)=e  
[1],e(2)=e[2],e(3)=s[pd],%):  
Lateral_Magnification:=normal(%);  
1/DD(m,m+3):  
subs(m=0,phi(1)=phi[1],phi(2)=phi[2],phi(3)=phi[3],e(0)=-  
s[p],e(1)=e[1],e(2)=e[2],%):  
Lateral_Magnification:=normal(%);
```

Lateral\_Magnification := 1

Lateral\_Magnification := 1

### Lateral magnification

#### Lateral magnification by means of axial distances measured from the principal points

[ 主点位置を基準として結像関係の横倍率を求める。

```
> LateralMagnification[PrincipalPoints]:=-(-s[pd]+s[kd_RayTrac  
e])/(s[p]+e[0]);
```

$$\text{LateralMagnification}_{\text{PrincipalPoints}} := \frac{-1400814718442685954598961}{14870687693452272826135607}$$

#### Check

[ 主点位置を基準とした横倍率が各面系で求めたものに等しいことを確認する。

```
> LateralMagnification[PrincipalPoints]-LateralMagnification[R  
ayTrace_Height]:  
normal(%);
```

0

### Imaging relation : $1/b-1/a = 1/f'$ & Newton equation : $zz' = ff'$

[ Gaussの結像公式及びNewton公式で結像関係を考察する。

### Image space

[ 像空間を考察する.

## Focal length (Image space)

### Focal length

[ 像空間の焦点距離を求める.

#### Focal length by means of paraxial ray tracing

[ 近軸光線追跡で焦点距離を求める.

```
> alpha[0]:=0; h[1]:=1;
alpha[1]:=alpha[0]+h[1]*phi[1]:
h[2]:=h[1]-alpha[1]*e[1]:
alpha[2]:=alpha[1]+h[2]*phi[2]:
h[3]:=h[2]-alpha[2]*e[2]:
alpha[3]:=alpha[2]+h[3]*phi[3]:
alpha_0 := 0
h_1 := 1
> f[d_RayTrace]:=h[1]/alpha[3];
f_d_RayTrace :=  $\frac{1110099903700545133890491}{4516311713281087658566}$ 
```

#### Focal length by means of Gaussian brackets

[ ABCDガウス定数で焦点距離を求める.

```
> 1/CC(m,m+2):
f[d_GaussianBrackets]:=subs(m=1,phi(1)=phi[1],
phi(2)=phi[2],phi(3)=phi[3],
e(1)=e[1],e(2)=e[2],%);
f_d_GaussianBrackets :=  $\frac{1110099903700545133890491}{4516311713281087658566}$ 
```

### Check

[ 両方法で求めた焦点距離が等しいことを確認する.

```
> f[d_RayTrace]-f[d_GaussianBrackets];
0
```

## Back focus

[ バックフォーカス(最終面から後側焦点までの距離)を2つの方法で求める.

### Back focus

#### Back focus by means of paraxial ray tracing

[ 初期値を  $\alpha'_0 = 0, h_1 = 1$  として、近軸光線追跡でバックフォーカスを求める.

```
> alpha[0]:=0; h[1]:=1;
alpha[1]:=alpha[0]+h[1]*phi[1]:
h[2]:=h[1]-alpha[1]*e[1]:
alpha[2]:=alpha[1]+h[2]*phi[2]:
h[3]:=h[2]-alpha[2]*e[2]:
alpha[3]:=alpha[2]+h[3]*phi[3]:
alpha_0 := 0
h_1 := 1
> s[Fd_RayTrace]:=h[3]/alpha[3];
s_Fd_RayTrace :=  $\frac{1069378362868878424032035}{4516311713281087658566}$ 
```

#### Back focus by means of Gaussian brackets

[ ABCDガウス係数によりバックフォーカスを求める.

```
> AA(m,m+2)/CC(m,m+2):  
s[Fd_GaussianBrackets]:=subs(m=1,phi(1)=phi[1],phi(2)=  
phi[2],phi(3)=phi[3],  
e(1)=e[1],e(2)=e[2],%);
```

$$s_{Fd\_GaussianBrackets} := \frac{1069378362868878424032035}{4516311713281087658566}$$

#### Check

[ 両方法で求めたバックフォーカスが等しいことを確認する.

```
> s[Fd_RayTrace]-s[Fd_GaussianBrackets];  
0
```

#### Relation among s[pd], s[Fd] and f[d]

[ 焦点距離、バックフォーカス、及び後側主点位置の関係を確認する.

```
> -s[pd]+s[Fd_RayTrace]-f[d_RayTrace]: normal(%);  
0
```

#### Object space

[ 次に、物空間を考察する.

#### Focal length (Object space)

[ 物空間の焦点距離を求める.

```
> -1/CC(m,m+2):  
f[_GaussianBrackets]:=subs(m=1,phi(1)=phi[1],phi(2)=phi[2],p  
hi(3)=phi[3],e(1)=e[1],e(2)=e[2],%);
```

$$f_{\_GaussianBrackets} := \frac{-1110099903700545133890491}{4516311713281087658566}$$

#### Front focus

[ フロントフォーカス(第1面より前側焦点までの距離)をABCDガウス係数により求める.

```
> -DD(m,m+2)/CC(m,m+2):  
s[F_GaussianBrackets]:=subs(m=1,phi(1)=phi[1],phi(2)=phi[2],  
phi(3)=phi[3],e(1)=e[1],e(2)=e[2],%);
```

$$s_{F\_GaussianBrackets} := \frac{-1125025184242537493789105}{4516311713281087658566}$$

#### Relation among s[p], s[F] and f[ ]

[ 焦点距離、フロントフォーカス、及び前側主点位置の関係を確認する.

```
> -s[p]+s[F_GaussianBrackets]-f[_GaussianBrackets]: normal(%);  
0
```

#### Gaussian imaging relation : $1/b-1/a=1/f'$ & Newton equation : $zz'=ff'$

#### $1/b-1/a=1/f'$

$1/b-1/a = 1/f'$  を確認する. ただし、 $a$ は前側主点から物点まで、 $b$ は後側主点から像点までの距離、そして $f'$ は焦点距離.

```
> 1/(s[kd_RayTrace]-s[pd])-1/(-(e[0]+s[p]))-1/f[d_RayTrace]:  
normal(%);
```

0

#### Newton equation : $zz'=ff'$

$zz' = f f'$  を確認する.  $z$ は前側焦点から物点まで、 $z'$ は後側焦点から像点までの距離、そして $f'$ は焦点距離.

```
> (-f[_GaussianBrackets]-s[p]-e[0])*(-f[d_GaussianBr  
ackets]-s[pd]+s[kd_RayTrace])
```

```
-f[_GaussianBrackets]*f[d_GaussianBrackets]:
```



```
normal(%);
```

0

## Imaging based on the nodal points

ここで、節点位置を基準とした結像を考察する。節点は角倍率が+1である共役点である。物体頂点から出て前側節点へ向かう光線は後側節点より出て像頂点へ向かい、かつ、入射光線と射出光線の傾角(換算傾角ではなく、実傾角)は等しい。従って、横倍率は幾何学的な考察、三角形の相似関係、より次式で与えられる。

$$\text{Lateral magnification} = -\frac{(s'_K - s'_N)n'_K}{(e'_0 + s_N)n'_0}$$

## Nodal points

### Positions of nodal points (reduced values)

まず、節点位置(換算量)を求める。  
前側節点位置(第1面より前側節点までの距離)及び後側節点位置(最終面より後側節点までの距離)をそれぞれABCDガウス定数を用いて求める。

```
> s[n]:= (n[3]/n[0]-d[GaussianBrackets])/c[GaussianBrackets];  
s[nd]:= (a[GaussianBrackets]-n[0]/n[3])/c[GaussianBrackets];
```

$$s_n := \frac{-102857168874435855446891242}{2937860769489347521897183}$$

$$s_{nd} := \frac{-113740723243541977969412417}{2504294845014363106674847}$$

### Check

#### Conjugate relation

前側節点を物体、後側節点を像とする結像関係を確認する。

```
> BB(m,m+4):  
subs(m=0,phi(1)=phi[1],phi(2)=phi[2],phi(3)=phi[3],e(0)=-s[n],  
e(1)=e[1],e(2)=e[2],e(3)=s[nd],%):  
normal(%);
```

0

#### Angular magnification is unity

角倍率が+1となることを確認する。

```
> (n[0]/n[3])*DD(m,m+3):  
subs(m=0,e(0)=-s[n],e(1)=e[1],e(2)=e[2],phi(1)=phi[1],phi(2)  
=phi[2],phi(3)=phi[3],%):  
Angular_Magnification:=normal(%);  
1*(n[0]/n[3])/AA(m,m+3):  
subs(m=1,e(1)=e[1],e(2)=e[2],e(3)=s[nd],phi(1)=phi[1],phi(2)  
=phi[2],phi(3)=phi[3],%):  
Angular_Magnification:=normal(%);
```

$$\text{Angular\_Magnification} := 1$$

$$\text{Angular\_Magnification} := 1$$

## Lateral magnification

節点位置を基準として結像関係の横倍率を求める。

### Nodal points (actual values)

節点位置(実寸法)を求める。

```
> S[N]:=s[n]*n[0];  
S[ND]:=s[nd]*n[3];
```

$$S_N := \frac{-99764470295282110035782}{2258155856640543829283}$$

$$S_{ND} := \frac{-11272618755542099077713}{2258155856640543829283}$$

## Lateral magnification

### Lateral magnification by means of axial distances measured from the nodal points (Actual values)

[ 節点位置を基準として結像関係の横倍率を求める.

> `LateralMagnification[NodalPoints]:=-(-S[ND]+s[kd_RayTrace]*n[3])/(S[N]+e[0]*n[0]);`

$$\text{LateralMagnification}_{\text{NodalPoints}} := \frac{-1400814718442685954598961}{14870687693452272826135607}$$

### Check

[ 節点位置を基準とした横倍率が各面系で求めたものに等しいことを確認する.

> `LateralMagnification[NodalPoints]-LateralMagnification[RayTrace_Height]:  
normal(%);`

0

## References

1. M.Herzberger, "Modern Geometrical Optics" (Interscience Publishers, New York, 1958) pp.457-462.
2. M.Herzberger, J.Opt.Soc.Amer. 33 No.12 (1943) 651, *ibid* 42 No.9 (1952) 637.
3. 宮本健郎, "光学入門" (岩波書店,東京,1995) pp.66-77.
4. O.Perron, "Die Lehre von den Kettenbruchen" (Teubner, Leipzig, 1913).
5. K.Tanaka, "Paraxial Theory in Optical Design in Terms of Gaussian Brackets" in *Progress in Optics*, vol.XXIII, ed.E.Wolf, (Elsevier, Amsterdam, 1986) pp.63-111.
6. K.Tanaka, "Paraxial Theory of Mechanically Compensated Zoom Lenses by Means of Gaussian Brackets", Canon Research Report No.6 (Canon Inc., Tokyo, 1991).
7. K.Tanaka, OPTIK 58 (1981) 351, *ibid* 62 (1982) 211, *ibid* 64 (1983) 13, *ibid* 64 (1983) 345, *ibid* 72 (1986) 125, *ibid* 72 (1986) 149, *ibid* 83 (1989) 19, *ibid* 97(1994) 129, J.Opt.(Paris) 16 (1985) 251, Appl.Opt. 23 (1984) 1700, 3265 (errata), *ibid* 25 (1986) 1531, Proc.SPIE 655 (1986) 44, *ibid* 813 (1987) 567.
8. E.W.Weisstein, "Concise Encyclopedia of Mathematics CD-ROM" Chapman & Hall/CRCnetBASE, 1999).
9. E.W.Weisstein, "Eric Weisstein's World of Mathematics The Web's most extensive Mathematics Resource" <http://mathworld.wolfram.com/GaussianBrackets.html>